
    Blink

Blink is a fast bulk data insertion library for Ecto and PostgreSQL.
It provides a callback-based pattern for seeding databases with dependent tables and shared context. Designed for scenarios where you need to insert test data, seed development databases, or populate staging environments quickly.
Installation
Add blink to your list of dependencies in mix.exs:
def deps do
  [
    {:blink, "~> 0.2.0"}
  ]
end
Then run mix deps.get to install it.
Example
defmodule MyApp.Seeder do
  use Blink

  def call do
    new()
    |> add_table("users")
    |> add_table("posts")
    |> insert(MyApp.Repo)
  end

  def table(_store, "users") do
    for i <- 1..1000 do
      %{
        id: i,
        name: "User #{i}",
        email: "user#{i}@example.com",
        inserted_at: DateTime.utc_now(),
        updated_at: DateTime.utc_now()
      }
    end
  end

  def table(store, "posts") do
    users = store.tables["users"]

    Enum.flat_map(users, fn user ->
      for i <- 1..5 do
        %{
          id: (user.id - 1) * 5 + i,
          title: "Post #{i}",
          user_id: user.id,
          inserted_at: DateTime.utc_now(),
          updated_at: DateTime.utc_now()
        }
      end
    end)
  end
end

# Inserts 1,000 users and 5,000 posts
MyApp.Seeder.call()
Features
	Fast bulk inserts - Uses PostgreSQL's COPY FROM STDIN command for optimal performance
	Dependent tables - Insert tables in order with access to previously inserted data
	Shared context - Compute expensive operations once and share across tables
	File loading - Built-in helpers for CSV and JSON file imports
	Configurable batching - Adjust batch sizes for memory-efficient large dataset insertion
	Transaction support - Automatic rollback on errors

Usage
Blink uses a callback-based pattern where you define:
	Which tables to insert (via add_table/2)
	What data goes in each table (via table/2 callback)
	Optional shared context (via add_context/2 and context/2 callback)

Accessing Previously Inserted Tables
Tables are inserted in the order they're added. Access previous table data via store.tables:
def table(store, "posts") do
  users = store.tables["users"]  # Access users inserted earlier

  Enum.flat_map(users, fn user ->
    for i <- 1..3 do
      %{
        id: (user.id - 1) * 3 + i,
        title: "Post #{i}",
        user_id: user.id,
        inserted_at: DateTime.utc_now(),
        updated_at: DateTime.utc_now()
      }
    end
  end)
end
Using Data from Context
Use context to compute expensive operations once and share across all tables:
def call do
  new()
  |> add_context("timestamps")
  |> add_table("users")
  |> add_table("posts")
  |> insert(MyApp.Repo)
end

def context(_store, "timestamps") do
  base = ~U[2024-01-01 00:00:00Z]
  for day <- 0..29, do: DateTime.add(base, day, :day)
end

def table(store, "users") do
  timestamps = store.context["timestamps"]
  # Use shared timestamps...
end

def table(store, "posts") do
  timestamps = store.context["timestamps"]
  # Reuse same timestamps...
end
Loading from Files
Load data from CSV or JSON files:
def table(_store, "users") do
  Blink.from_csv("priv/seed_data/users.csv",
    transform: fn row ->
      row
      |> Map.update!("id", &String.to_integer/1)
      |> Map.put("inserted_at", DateTime.utc_now())
      |> Map.put("updated_at", DateTime.utc_now())
    end
  )
end

def table(_store, "products") do
  Blink.from_json("priv/seed_data/products.json",
    transform: fn product ->
      Map.put(product, "inserted_at", DateTime.utc_now())
    end
  )
end
CSV files use the first row as headers by default. Both helpers accept a :transform option for type conversion or data manipulation.
Configuring Batch Size
Adjust batch size for large datasets:
new()
|> add_table("users")
|> insert(MyApp.Repo, batch_size: 5_000)  # Default: 900
Using with ExMachina
Combine ExMachina's factory pattern with Blink's fast insertion:
defmodule MyApp.Seeder do
  use Blink
  import MyApp.Factory

  def call do
    new()
    |> add_table("users")
    |> add_table("posts")
    |> insert(MyApp.Repo)
  end

  def table(_store, "users") do
    for _i <- 1..100 do
      user = build(:user)
      Map.put(user, :id, Ecto.UUID.generate())
    end
  end

  def table(store, "posts") do
    user_ids = Enum.map(store.tables["users"], & &1.id)

    Enum.flat_map(user_ids, fn user_id ->
      for _i <- 1..5 do
        post = build(:post, user_id: user_id)
        Map.put(post, :id, Ecto.UUID.generate())
      end
    end)
  end
end
Learning Blink
	Getting Started guide - Step-by-step tutorial building a complete seeding system
	API documentation - Full reference for all functions and callbacks
	Changelog - Version history and migration guides

Requirements
	Requirement	Version
	Elixir	1.14+
	Ecto	3.0+
	PostgreSQL	Any supported version

Known Limitations
Memory usage with large datasets - Blink loads all table data into memory before insertion. For very large datasets, consider splitting your seeder into multiple modules:
# Instead of one large seeder, use multiple smaller ones
organization_ids = OrganizationSeeder.call()
user_ids = UserSeeder.call(organization_ids)
PostSeeder.call(user_ids)
This limitation may be addressed in a future version.
License
Copyright (c) 2026 Nerds and Company
Licensed under the MIT License. See LICENSE for details.


  

    Changelog

All notable changes to this project will be documented in this file.
The format is based on Keep a Changelog,
and this project adheres to Semantic Versioning.
[Unreleased]
0.2.0 - 2026-01-09
Added
	Blink.from_csv/2 function for reading CSV files into maps
	Support for CSV files with headers (inferred from first row by default)
	Support for CSV files without headers via :headers option
	:transform option for CSV type conversion and data transformation
	Blink.from_json/2 function for reading JSON files into maps
	Support for JSON arrays of objects with automatic type preservation
	:transform option for JSON data transformation
	New guide: "Loading Data from Files"
	New guide: "Using Context"
	New guide: "Integrating with ExMachina"

Changed
	Simplified "Getting Started" guide to focus on core concepts
	Refactored CSV and JSON functionality into dedicated internal modules

0.1.1 - 2026-01-08
Changed
	Lowered Elixir requirement from 1.18 to 1.14 for better compatibility
	Improved package description and documentation

0.1.0 - 2026-01-08
Added
	Initial release of Blink
	Fast bulk data insertion using PostgreSQL's COPY command
	Callback-based pattern for defining seeders with use Blink
	Support for multiple tables with insertion order
	Context sharing between table definitions
	Configurable batch size for large datasets
	Transaction support with automatic rollback on errors
	Comprehensive test suite with integration tests
	Full documentation and examples



  

    LICENSE


MIT License

Copyright (c) 2026 Nerds & Company

Permission is hereby granted, free of charge, to any person obtaining a copy
of this software and associated documentation files (the "Software"), to deal
in the Software without restriction, including without limitation the rights
to use, copy, modify, merge, publish, distribute, sublicense, and/or sell
copies of the Software, and to permit persons to whom the Software is
furnished to do so, subject to the following conditions:

The above copyright notice and this permission notice shall be included in all
copies or substantial portions of the Software.

THE SOFTWARE IS PROVIDED "AS IS", WITHOUT WARRANTY OF ANY KIND, EXPRESS OR
IMPLIED, INCLUDING BUT NOT LIMITED TO THE WARRANTIES OF MERCHANTABILITY,
FITNESS FOR A PARTICULAR PURPOSE AND NONINFRINGEMENT. IN NO EVENT SHALL THE
AUTHORS OR COPYRIGHT HOLDERS BE LIABLE FOR ANY CLAIM, DAMAGES OR OTHER
LIABILITY, WHETHER IN AN ACTION OF CONTRACT, TORT OR OTHERWISE, ARISING FROM,
OUT OF OR IN CONNECTION WITH THE SOFTWARE OR THE USE OR OTHER DEALINGS IN THE
SOFTWARE.



  

    Getting Started

This guide is an introduction to Blink, a fast bulk data insertion library for Ecto and PostgreSQL.
In this guide, we are going to:
	Create a seeder module for inserting users and posts
	Learn how to access data from previously inserted tables
	Configure batch sizes for optimal performance

Adding Blink to an application
Add Blink to your dependencies in mix.exs:
defp deps do
  [
    {:blink, "~> 0.2.0"}
  ]
end
Install the dependencies:
mix deps.get

Configuring the repository
Blink works with any Ecto repository. If you don't have Ecto set up yet, follow the Ecto Getting Started guide to configure your repository and create your database tables.
For this guide, we'll assume you have:
	An Ecto repository (e.g., Blog.Repo) configured
	A users table with columns: id, name, email, inserted_at, updated_at
	A posts table with columns: id, title, body, user_id, inserted_at, updated_at

Creating a seeder
Now that we have our database set up, let's create a seeder to insert data. Create lib/blog/seeders/blog_seeder.ex:
defmodule Blog.Seeders.BlogSeeder do
  use Blink

  def call do
    new()
    |> add_table("users")
    |> insert(Blog.Repo)
  end

  def table(_store, "users") do
    for i <- 1..100 do
      %{
        id: i,
        name: "User #{i}",
        email: "user#{i}@example.com",
        inserted_at: ~U[2024-01-01 00:00:00Z],
        updated_at: ~U[2024-01-01 00:00:00Z]
      }
    end
  end
end
The seeder above does the following:
	use Blink - Imports Blink's functions and defines required callbacks
	new() - Creates an empty container, called a store, to hold our table data
	add_table("users") - Registers the users table for insertion
	table/2 callback - Defines what data to insert into the users table
	insert/2 - Executes the bulk insertion using PostgreSQL's COPY command

Let's run it from IEx:
iex -S mix
iex> Blog.Seeders.BlogSeeder.call()
# => Inserts 100 users
Inserting dependent tables
Most applications have relationships between tables. Let's add posts that belong to users. Update the seeder:
def call do
  new()
  |> add_table("users")
  |> add_table("posts")  # Add the posts table
  |> insert(Blog.Repo)
end

# ... existing table/2 for users ...

def table(store, "posts") do
  users = store.tables["users"]  # Access previously inserted users

  Enum.flat_map(users, fn user ->
    for i <- 1..5 do
      %{
        id: (user.id - 1) * 5 + i,
        title: "Post #{i} by #{user.name}",
        body: "This is the content of post #{i}",
        user_id: user.id,
        inserted_at: ~U[2024-01-01 00:00:00Z],
        updated_at: ~U[2024-01-01 00:00:00Z]
      }
    end
  end)
end
The key insight here is that tables are inserted in the order they're added. When defining the "posts" table, we can access the "users" table data via store.tables["users"]. This allows us to reference user IDs when creating posts.
Run the updated seeder:
iex> Blog.Seeders.BlogSeeder.call()
# => Inserts 100 users and 500 posts
Configuring batch size
By default, Blink inserts records in batches of 900. You can configure this for optimal performance:
def call do
  new()
  |> add_table("users")
  |> insert(Blog.Repo, batch_size: 1_200)
end
A larger batch size can improve performance for large datasets, while a smaller batch size may be more suitable for records with many columns or large text fields.
Summary
In this guide, we learned how to:
	Create a seeder module with use Blink
	Insert data into multiple related tables
	Access previously inserted table data via store.tables
	Configure batch sizes for performance

Next steps
You might also find these guides useful:
	Using Context - Share computed data across tables
	Loading Data from Files - Learn how to load data from CSV and JSON files
	Integrating with ExMachina - Generate realistic test data

For more information, see the Blink API documentation.


  

    Using Context

Sometimes you need to compute data once and share it across multiple tables. Blink provides the context feature for this purpose. Context data is not inserted into the database, but is available when building your table data.
What is context?
Context is arbitrary data stored in store.context that you can access from any table/2 or context/2 callback. It's useful for:
	Sharing computed values across tables (e.g., timestamps, IDs)
	Pre-generating data that multiple tables need
	Storing lookup tables or reference data
	Avoiding redundant computations

Basic example
Let's say we want to generate consistent timestamps and use them across multiple tables:
defmodule Blog.Seeders.BlogSeeder do
  use Blink

  def call do
    new()
    |> add_context("timestamps")  # Register context first
    |> add_table("users")
    |> add_table("posts")
    |> insert(Blog.Repo)
  end

  def context(_store, "timestamps") do
    # Generate 30 days of timestamps
    base = ~U[2024-01-01 00:00:00Z]
    for day <- 0..29, do: DateTime.add(base, day, :day)
  end

  def table(store, "users") do
    timestamps = store.context["timestamps"]

    for i <- 1..100 do
      %{
        id: i,
        name: "User #{i}",
        email: "user#{i}@example.com",
        inserted_at: Enum.random(timestamps),
        updated_at: Enum.random(timestamps)
      }
    end
  end

  def table(store, "posts") do
    users = store.tables["users"]
    timestamps = store.context["timestamps"]

    Enum.flat_map(users, fn user ->
      for i <- 1..5 do
        %{
          id: (user.id - 1) * 5 + i,
          title: "Post #{i}",
          body: "Content here",
          user_id: user.id,
          inserted_at: Enum.random(timestamps),
          updated_at: Enum.random(timestamps)
        }
      end
    end)
  end
end
In this example, we generate timestamps once and reuse them across both the users and posts tables.
Context with relationships
Context can help maintain referential integrity by providing consistent reference data:
def call do
  new()
  |> add_context("user_ids")
  |> add_table("users")
  |> add_table("posts")
  |> add_table("comments")
  |> insert(Blog.Repo)
end

def context(_store, "user_ids") do
  # Generate a pool of user IDs
  Enum.to_list(1..1000)
end

def table(store, "users") do
  user_ids = store.context["user_ids"]

  for id <- user_ids do
    %{
      id: id,
      name: "User #{id}",
      email: "user#{id}@example.com",
      inserted_at: ~U[2024-01-01 00:00:00Z],
      updated_at: ~U[2024-01-01 00:00:00Z]
    }
  end
end

def table(store, "posts") do
  user_ids = store.context["user_ids"]

  for i <- 1..5000 do
    %{
      id: i,
      title: "Post #{i}",
      body: "Content",
      user_id: Enum.random(user_ids),
      inserted_at: ~U[2024-01-01 00:00:00Z],
      updated_at: ~U[2024-01-01 00:00:00Z]
    }
  end
end

def table(store, "comments") do
  user_ids = store.context["user_ids"]
  posts = store.tables["posts"]

  Enum.flat_map(posts, fn post ->
    for i <- 1..3 do
      %{
        id: (post.id - 1) * 3 + i,
        body: "Comment #{i}",
        post_id: post.id,
        user_id: Enum.random(user_ids),
        inserted_at: ~U[2024-01-01 00:00:00Z],
        updated_at: ~U[2024-01-01 00:00:00Z]
      }
    end
  end)
end
Context for realistic data
Use context to generate realistic, consistent data:
def call do
  new()
  |> add_context("timestamps")
  |> add_table("users")
  |> add_table("posts")
  |> insert(Blog.Repo)
end

def context(_store, "timestamps") do
  base = ~U[2024-01-01 00:00:00Z]
  for day <- 0..29, do: DateTime.add(base, day, :day)
end

def table(store, "users") do
  timestamps = store.context["timestamps"]

  for i <- 1..100 do
    %{
      id: i,
      name: "User #{i}",
      email: "user#{i}@example.com",
      inserted_at: Enum.random(timestamps),
      updated_at: Enum.random(timestamps)
    }
  end
end

def table(store, "posts") do
  users = store.tables["users"]
  timestamps = store.context["timestamps"]

  Enum.flat_map(users, fn user ->
    # Only use timestamps after the user was created
    valid_timestamps =
      Enum.filter(timestamps, fn ts ->
        DateTime.compare(ts, user.inserted_at) == :gt
      end)

    for i <- 1..5 do
      %{
        id: (user.id - 1) * 5 + i,
        title: "Post #{i}",
        body: "Content here",
        user_id: user.id,
        inserted_at: Enum.random(valid_timestamps),
        updated_at: Enum.random(valid_timestamps)
      }
    end
  end)
end
In this example, we ensure posts are created after their associated user by filtering the available timestamps.
Summary
In this guide, we learned how to:
	Add context data with add_context/2
	Define context callbacks with context/2
	Access context from table callbacks via store.context

For more information, see the Blink API documentation.


  

    Loading Data from Files

Blink provides helper functions to load data from CSV and JSON files, making it easy to seed your database from external data sources.
Loading from CSV files
CSV files are a common format for storing tabular data. Blink can read CSV files and convert them into maps suitable for insertion.
Basic usage
Create a CSV file at priv/seed_data/users.csv:
id,name,email
1,Alice Johnson,alice@example.com
2,Bob Smith,bob@example.com
3,Carol White,carol@example.com
Load it in your seeder:
defmodule Blog.Seeders.BlogSeeder do
  use Blink

  def call do
    new()
    |> add_table("users")
    |> insert(Blog.Repo)
  end

  def table(_store, "users") do
    Blink.from_csv("priv/seed_data/users.csv")
  end
end
By default, from_csv/2 reads the first row as column headers and returns a list of maps with string keys. All values are returned as strings.
Transforming data
Use the :transform option to convert types and add required fields:
def table(_store, "users") do
  base_time = ~U[2024-01-01 00:00:00Z]

  Blink.from_csv("priv/seed_data/users.csv",
    transform: fn row ->
      row
      |> Map.update!("id", &String.to_integer/1)
      |> Map.put("inserted_at", base_time)
      |> Map.put("updated_at", base_time)
    end
  )
end
The transform function receives each row as a map and should return the transformed map.
CSV files without headers
If your CSV file doesn't have a header row, provide the column names explicitly:
def table(_store, "users") do
  Blink.from_csv("priv/seed_data/users_no_headers.csv",
    headers: ["id", "name", "email"]
  )
end
Combining headers and transform
You can use both options together:
def table(_store, "users") do
  Blink.from_csv("priv/seed_data/users_no_headers.csv",
    headers: ["id", "name", "email"],
    transform: fn row ->
      Map.update!(row, "id", &String.to_integer/1)
    end
  )
end
Loading from JSON files
JSON files are useful when your data includes nested structures or when you need to preserve data types.
Basic usage
Create a JSON file at priv/seed_data/products.json:
[
  {"id": 1, "name": "Widget", "price": 9.99},
  {"id": 2, "name": "Gadget", "price": 19.99},
  {"id": 3, "name": "Doohickey", "price": 29.99}
]
Load it in your seeder:
def table(_store, "products") do
  Blink.from_json("priv/seed_data/products.json")
end
The JSON file must contain an array of objects at the root level. Each object becomes a map with string keys.
Transforming JSON data
Use the :transform option to add timestamps or modify fields:
def table(_store, "products") do
  Blink.from_json("priv/seed_data/products.json",
    transform: fn product ->
      Map.merge(product, %{
        "inserted_at" => ~U[2024-01-01 00:00:00Z],
        "updated_at" => ~U[2024-01-01 00:00:00Z]
      })
    end
  )
end
Error handling
The functions from_csv/2 and from_json/2 will raise exceptions if:
	The file doesn't exist
	The file format is invalid
	The :transform function is not a single-arity function
	For JSON: the root element is not an array, or the array contains non-object elements
	For CSV: the :headers option is not :infer or a list of strings

These errors help catch issues early in your seeding process.
Summary
In this guide, we learned how to:
	Load data from CSV files with from_csv/2
	Load data from JSON files with from_json/2
	Transform data with the :transform option
	Handle CSV files without headers

For more information, see the Blink API documentation.


  

    Integrating with ExMachina

ExMachina is a popular library for generating test data in Elixir. Blink works seamlessly with ExMachina, allowing you to combine ExMachina's expressive factories with Blink's high-performance bulk insertion.
Why combine Blink and ExMachina?
	ExMachina: Generates realistic, varied test data with factories
	Blink: Performs fast bulk insertions into PostgreSQL

Together, they provide the best of both worlds: expressive data generation with efficient database seeding.
Setting up ExMachina
Add ExMachina to your dependencies in mix.exs:
defp deps do
  [
    {:ex_machina, "~> 2.7", only: [:dev, :test]}
  ]
end
Install the dependencies:
mix deps.get

Creating factories
Create a factory module:
defmodule Blog.Factory do
  use ExMachina

  def user_factory do
    %{
      name: Faker.Person.name(),
      email: Faker.Internet.email()
    }
  end

  def post_factory do
    %{
      title: Faker.Lorem.sentence(),
      body: Faker.Lorem.paragraph()
    }
  end
end
Note: To use Faker, add {:faker, "~> 0.18", only: [:dev, :test]} to your dependencies.
Basic integration
Use ExMachina's build/1 function in your Blink seeder:
defmodule Blog.Seeders.BlogSeeder do
  use Blink
  import Blog.Factory

  def call do
    new()
    |> add_table("users")
    |> insert(Blog.Repo)
  end

  def table(_store, "users") do
    for i <- 1..1000 do
      user = build(:user)

      Map.merge(user, %{
        id: i,
        inserted_at: ~U[2024-01-01 00:00:00Z],
        updated_at: ~U[2024-01-01 00:00:00Z]
      })
    end
  end
end
ExMachina generates the names and emails, while you control the IDs and timestamps.
Using sequences
ExMachina provides sequences for unique values:
defmodule Blog.Factory do
  use ExMachina

  def user_factory do
    %{
      name: Faker.Person.name(),
      email: sequence(:email, &"user#{&1}@example.com"),
      username: sequence(:username, &"user#{&1}")
    }
  end
end
Use them in your seeder:
def table(_store, "users") do
  for i <- 1..1000 do
    user = build(:user)

    Map.merge(user, %{
      id: i,
      inserted_at: ~U[2024-01-01 00:00:00Z],
      updated_at: ~U[2024-01-01 00:00:00Z]
    })
  end
end
Each user will have a unique email and username generated by ExMachina's sequences.
Summary
In this guide, we learned how to:
	Set up ExMachina for data generation
	Use ExMachina's build/1 function with Blink seeders
	Work with sequences for unique values

For more information:
	ExMachina documentation
	Blink API documentation

Blink provides an efficient way to seed large amounts of data into your
database.
Overview
Blink simplifies database seeding by providing a structured way to build and
insert records:
	Create an empty Store.
	Assign the records you want to insert to each database table.
	Bulk-insert the records into your database.

Stores
Stores are the central data unit in Blink. A Store is a struct that holds
the records you want to seed, along with any contextual data you need during
the seeding process but do not want to insert into the database.
A Store struct contains the keys tables and context:
Blink.Store{
  tables: %{
    "table_name" => [...]
  },
  context: %{
    "key" => [...]
  }
}
All keys in tables must match the name of a table in your database. Table
names can be either atoms or strings.
Tables
A mapping of table names to lists of records. These records will be persisted
to the database when insert/2 or insert/3 are called.
Context
Stores arbitrary data needed during the seeding process. This data is
available when building your seeds but is not inserted into the database by
insert/2 or insert/3.
Basic Usage
To seed your database with Blink, follow these three steps:
	Create: Initialize an empty store with new/0.

	Build: Add seed data with add_table/2 and context data with
add_context/2.

	Insert: Persist records to the database with insert/2 or insert/3.


Example
defmodule MyApp.Seeder do
  use Blink

  def call do
    new()
    |> add_table("users")
    |> add_context("post_ids")
    |> insert(MyApp.Repo, batch_size: 1_200)
  end

  def table(_store, "users") do
    [
      %{id: 1, name: "Alice", email: "alice@example.com"},
      %{id: 2, name: "Bob", email: "bob@example.com"}
    ]
  end

  def context(_store, "post_ids") do
    [1, 2, 3]
  end
end
Custom Logic for Inserting Records
The functions insert/2 and insert/3 bulk insert the table records in a
Store into a Postgres database using Postgres' COPY command. You can
override the default implementation by defining your own insert/2 or
insert/3 function in your Blink module. Doing so you can support seeding
databases other than Postgres.

      


      
        Summary


  
    Callbacks
  


    
      
        context(store, table_or_context_key)

      


        Builds and returns the data to be stored under a context key in the given
Store.



    


    
      
        insert(store, repo)

      


        Specifies how to perform a bulk insert of the seed data from a Store into
the given Ecto repository.



    


    
      
        insert(store, repo, opts)

      


    


    
      
        table(store, table_name)

      


        Builds and returns the records to be stored under a table key in the given
Store.



    





  
    Functions
  


    
      
        from_csv(path, opts \\ [])

      


        Reads a CSV file and returns a list of maps suitable for use in table/2 callbacks.



    


    
      
        from_json(path, opts \\ [])

      


        Reads a JSON file and returns a list of maps suitable for use in table/2 callbacks.



    





      


      
        Callbacks


        


  
    
      
    
    
      context(store, table_or_context_key)


        (optional)


        
          
        

    

  


  

      

          @callback context(store :: Blink.Store.t(), table_or_context_key :: binary() | atom()) ::
  [map()]


      


Builds and returns the data to be stored under a context key in the given
Store.
The callback context/2 is called by add_context/2 internally, passing the
given context key to context/2. Therefore, each key passed to a
add_context/2 clause must match a context/2 clause.
insert/2 and insert/3 ignore the :context data and only insert data from
:tables.
When the callback function is missing, an ArgumentError is raised.

  



  
    
      
    
    
      insert(store, repo)


        (optional)


        
          
        

    

  


  

      

          @callback insert(store :: Blink.Store.t(), repo :: Ecto.Repo.t()) :: :ok | :error


      


Specifies how to perform a bulk insert of the seed data from a Store into
the given Ecto repository.
This callback function is optional, since Blink ships with a default
implementation for Postgres databases.

  



  
    
      
    
    
      insert(store, repo, opts)


        (optional)


        
          
        

    

  


  

      

          @callback insert(store :: Blink.Store.t(), repo :: Ecto.Repo.t(), opts :: Keyword.t()) ::
  :ok | :error


      



  



  
    
      
    
    
      table(store, table_name)


        (optional)


        
          
        

    

  


  

      

          @callback table(store :: Blink.Store.t(), table_name :: binary() | atom()) :: [map()]


      


Builds and returns the records to be stored under a table key in the given
Store.
The callback table/2 is called by add_table/2 internally, passing the
given database table name to table/2. Therefore, each table name passed to a
add_table/2 clause must match a table/2 clause.
Data added to a store with table/2 is inserted into the corresponding
database table when calling insert/2 or insert/3.
When the callback function is missing, an ArgumentError is raised.

      from_csv(path, opts \\ [])



        
          
        

    

  


  

      

          @spec from_csv(path :: String.t(), opts :: Keyword.t()) :: [map()]


      


Reads a CSV file and returns a list of maps suitable for use in table/2 callbacks.
By default, the CSV file must have a header row. Each column header will become a
string key in the resulting maps. All values are returned as strings.
Parameters
	path - Path to the CSV file (relative or absolute)
	opts - Keyword list of options:	:headers - List of header names to use, or :infer to read from first row (default: :infer)
	:transform - Function to transform each row map (default: identity function)



Examples
# Simple usage with headers in first row
def table(_store, "users") do
  Blink.from_csv("users.csv")
end

# CSV without headers - provide them explicitly
def table(_store, "users") do
  Blink.from_csv("users.csv", headers: ["id", "name", "email"])
end

# With custom transformation for type conversion
def table(_store, "users") do
  Blink.from_csv("users.csv",
    transform: fn row ->
      row
      |> Map.update!("id", &String.to_integer/1)
      |> Map.update!("age", &String.to_integer/1)
    end
  )
end
Returns
A list of maps, where each map represents a row from the CSV file.

  



    

  
    
      
    
    
      from_json(path, opts \\ [])



        
          
        

    

  


  

      

          @spec from_json(path :: String.t(), opts :: Keyword.t()) :: [map()]


      


Reads a JSON file and returns a list of maps suitable for use in table/2 callbacks.
The JSON file must contain an array of objects at the root level. Each object
becomes a map with string keys.
Parameters
	path - Path to the JSON file
	opts - Keyword list of options:	:transform - Function to transform each row map (default: identity function)



Examples
# Simple usage
def table(_store, "users") do
  Blink.from_json("users.json")
end

# With custom transformation for type conversion
def table(_store, "users") do
  Blink.from_json("users.json",
    transform: fn row ->
      row
      |> Map.update!("id", &String.to_integer/1)
      |> Map.update!("age", &String.to_integer/1)
    end
  )
end
Returns
A list of maps, where each map represents an object from the JSON array.

  


        

      


  

    
Blink.CSVParser 
    




      
        Summary


  
    Functions
  


    
      
        dump_to_iodata(enumerable)

      


        Callback implementation for NimbleCSV.dump_to_iodata/1.



    


    
      
        dump_to_stream(enumerable)

      


        Callback implementation for NimbleCSV.dump_to_stream/1.



    


    
      
        parse_enumerable(enumerable, opts \\ [])

      


        Callback implementation for NimbleCSV.parse_enumerable/2.



    


    
      
        parse_stream(stream, opts \\ [])

      


        Callback implementation for NimbleCSV.parse_stream/2.



    


    
      
        parse_string(string, opts \\ [])

      


        Callback implementation for NimbleCSV.parse_string/2.



    


    
      
        to_line_stream(stream)

      


        Callback implementation for NimbleCSV.to_line_stream/1.

      dump_to_iodata(enumerable)



        
          
        

    

  


  

Callback implementation for NimbleCSV.dump_to_iodata/1.

  



  
    
      
    
    
      dump_to_stream(enumerable)



        
          
        

    

  


  

Callback implementation for NimbleCSV.dump_to_stream/1.

  



    

  
    
      
    
    
      parse_enumerable(enumerable, opts \\ [])



        
          
        

    

  


  

Callback implementation for NimbleCSV.parse_enumerable/2.

  



    

  
    
      
    
    
      parse_stream(stream, opts \\ [])



        
          
        

    

  


  

Callback implementation for NimbleCSV.parse_stream/2.

  



    

  
    
      
    
    
      parse_string(string, opts \\ [])



        
          
        

    

  


  

Callback implementation for NimbleCSV.parse_string/2.

  



  
    
      
    
    
      to_line_stream(stream)



        
          
        

    

  


  

Callback implementation for NimbleCSV.to_line_stream/1.

  


        

      


  

    
Blink.Store 
    



      
The central data structure used throughout the Blink seeding pipeline.
A Store holds:
	:tables — data that will be inserted into the database.
	:context — auxiliary data available while constructing the Store, and
will not be inserted into the database.

          @type t() :: %Blink.Store{context: map(), tables: map()}

